

```

hwalk=: 4 : 'x([randadd interp@])^(y) 0,(x osz 1)*randsn 1'
setseed 7^5                                as defined in Section 3.5
'pensize 2' plot 0.1 hwalk 8
'pensize 2' plot 0.9 hwalk 8

```

Create eleven such random walks corresponding to $H = 0, 0.1, 0.2, \dots, 1.0$.

3.8 Experiment: Forecasting via Best Analogs

In this section we experiment with some forecasts of data based upon matching given patterns with the best analogs that can be made to historical reference patterns. This is sometimes called fractal forecasting because it offers very intriguing possibilities for taking advantage of hidden chaotic/fractal behavior that appears random. We will consider best analog forecasting of the Henon map and the sunspot data. The general strategy we use follows references Balkin (1996) and Casdagli (1992) but we will take the prediction based upon the single best fit analog, rather than using regression on several close fits.

We first define a function `dist` that can be used to find the distance between points. That is, it computes the square root of the sum of the squares of the differences between the points. We use the Henon map from Section 2.2 in order to generate our first set of data.

```

dist=: %:@(+/@):*:@:-"1
1 2 3 dist 3 4 2          distance between two points
3
1 2 3 dist 3 5 5,:3 4 2  distance between a point and a list of points
4.12311 3

```

The basic strategy is to take the historical data set and divide it up into blocks of fixed length M called M -histories: they provide a reference list of M -histories whose future behaviors are known. In practice, we take windows of size $M+1$ and consider the first M points to be the “reference” pattern and the last point to be the next “result” in the sequence. Thus, below we take size $M+1$ infixes of the Henon data time series where we are using $M = 2$.

```

hen=: 3 : '({:y),1+(0.3*{:y)+_1.4* *{:y'
$hd=:{:|: 10}.hen^(i. 510) 0 0    Henon data
500
$ref_res=:3 ]\ 450{.hd            reference and result lists combined
448 3
$ref=:}: "1 ref_res              reference list
448 2
$res=:{: "1 ref_res              next result list
448
_4{.ref                          last 4 pairs in the reference set
1.04842 _0.669762
_0.669762 0.686513
0.686513 0.139251
0.139251 1.17881
_4{.res                          next terms in each sequence (the results)
0.686513 0.139251 1.17881 _0.903645

```

```
$fhd=:450}.hd
50
```

future Henon data (it is not in the reference set)

```
]X=: 2{.fhd
0.210438 0.666909
```

a starting pattern for our predictions

Grade up gives the indices of a list in increasing order. Thus the first element of the grade up gives the index of the *M*-history closest to our data.

```
/:13 2 53 1 77 99
3 1 0 2 4 5
```

grade up gives indices of increasing order

```
{./:13 2 53 1 77 99
3
```

index of the smallest item

```
3{13 2 53 1 77 99
1
```

the smallest item

We see below that the index of the closest pattern to *X* is 241 and when we compare that pattern to *X*, we see they are quite close. Looking at the data at index 241 in *res* is our prediction for the next point in the sequence.

```
{.@/: ref dist X
241
```

```
241{ref
0.209871 0.668865
```

```
X
0.210438 0.666909
```

```
241{res
0.436629
```




Figure 3.8.1 Best Analogs Forecasts for Henon Data

We define a function `fpred_step` which puts all these steps together. Note, it presumes that *ref* and *res* have been correctly defined. The result is the subsequent pattern using the prediction. This makes the function suited for iteration.

```
fpred_step=:3 : 0
({.y),({.@/:ref dist y){res
)
```

```
fpred_step X
0.666909 0.436629
```

```
fpred_step^(i.4) X
0.210438 0.666909
0.666909 0.436629
0.436629 0.933757
0.933757 _0.0896736
```

The following records the actual 30 steps of the Henon map and computes a 30 step forecast.

```
$a=: 30{.fhd
30
```

```
$b={."1 fpred_step^(i.30) X
30
```

```
require 'plot'
```

```
plot ((i.30) ; a ,: b)
```

The following plot driver commands show a plot of the actual Henon data in blue, and the forecasts using `fpred_step` in red. The result is shown in Figure 3.8.1.

```
pd 'reset;'
pd 'type line'
pd 'pensize 5'
pd 'itemcolor 0 0 255'
pd a
pd 'itemcolor 255 0 0'
pd 'pensize 3'
pd b
pd 'show'
```




Figure 3.8.2 Best Analogs Forecasts for Sunspot Data

Next we repeat the same experiment for our sunspots data from `ts_data.ijs`. This time we use the nearest point to 11-histories for our estimates and attempt a 40 year forecast.

```
load '~addons/graphics/fvj4/ts_data.ijs'

$ref_res=:12 ]\ 350{.sunspots
339 12

$ref=:}:"1 ref_res
339 11

$res=:{"1 ref_res
339

fsd=:350{.sunspots

]X=: 11{.fsd
66.6 64.5 54.1 39 20.6 6.7 4.3 22.7 54.8 93.8 95.8

$a=: 40{.fsd
40

$b={."1 fpred_step^(i.40) X
40

plot ((i.40) ; a ,: b)
```

The result is shown in Figure 3.8.2.